The Stata Journal (2014) 14, Number 1, pp. 1–3

Stata tip 116: Where did my p-values go? (part 3)

Maarten L. Buis
Wissenschaftszentrum Berlin für Sozialforschung (WZB)
Berlin, Germany
maarten.buis@wzb.eu

In a previous Stata tip (Buis 2007) I discussed how to recover t-statistics, p-values, and confidence intervals for regression parameters using the results that are returned by an estimation command. In a subsequent Stata tip (Buis 2011) I discussed how to recover parameter estimates for parameters that were estimated on a transformed scale, for example if a likelihood function contains a standard deviation or a correlation then many Stata commands will maximize the likelihood with respect to ln(standard deviation) and the Fisher's z-transformation of the correlation. In this tip I will discuss how to recover the standard errors for the back-transformed parameters, that is, the standard errors of the standard deviation and the correlation.

Often Stata does display the back-transformed parameters and their standard errors, but it leaves behind only the estimates of the transformed parameters and their standard errors. In those cases the delta method (for example: Feiveson 2005) was used to compute the standard errors of the back-transformed parameters. In its simplest form the delta method means that if we apply a transformation $G(\cdot)$ to a parameter b, than we can approximate the standard error of the transformed parameter as:

$$se(G(b)) \approx se(b) \times G'(\hat{b})$$

where $G'(\hat{b})$ is the first derivative of $G(b)$ with respect to b evaluated at $\hat{b}$. If Stata returned ln(standard deviation) and we want the standard deviation and its standard error, then $G(b) = \exp(b)$ and $G'(\hat{b}) = \exp(\hat{b})$. If Stata returned Fisher's z-transformation of a correlation and we wanted the correlation and its standard error, then $G(b) = \tanh(b)$ and $G'(\hat{b}) = \cosh(\hat{b})^{-2}$. This is illustrated below using a model estimated with [R] **heckman**. This model was chosen because it returns transformed parameters of both types.

```
. webuse womenwk, clear
. heckman wage educ, select(married children educ) nolog
Heckman selection model                         Number of obs      =      2000
(regression model with sample selection)        Censored obs       =       657
                                                Uncensored obs     =      1343

                                                Wald chi2(1)       =    403.39
Log likelihood = -5250.348                      Prob > chi2        =    0.0000

------------------------------------------------------------------------------
        wage |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
wage         |
   education |   1.099506   .0547435    20.08   0.000     .9922102    1.206801
       _cons |   7.042147   .8423253     8.36   0.000      5.39122    8.693074
-------------+----------------------------------------------------------------
```

 st0001

```
select       |
     married |   .5420304   .0657798     8.24   0.000     .4131044    .6709564
    children |   .4409418   .0276093    15.97   0.000     .3868286     .495055
   education |   .0722993   .0105096     6.88   0.000     .0517007    .0928978
       _cons |  -1.473038   .1465476   -10.05   0.000    -1.760266    -1.18581
-------------+----------------------------------------------------------------
     /athrho |   .8081049   .1108545     7.29   0.000     .5908341    1.025376
    /lnsigma |   1.807547   .0291035    62.11   0.000     1.750506    1.864589
-------------+----------------------------------------------------------------
         rho |   .6685435    .061308                      .5304953     .772047
       sigma |   6.095479   .1773995                      5.757513    6.453283
      lambda |   4.075093   .4690025                      3.155865    4.994321
------------------------------------------------------------------------------
LR test of indep. eqns. (rho = 0):   chi2(1) =    47.02   Prob > chi2 = 0.0000

.
. tempname gprime
. scalar `gprime´  = exp(_b[lnsigma:_cons])
. di "se of sigma = " _se[lnsigma:_cons]*`gprime´
se of sigma = .17739954
.
. scalar `gprime´  = cosh(_b[athrho:_cons])^-2
. di "se of rho = " _se[athrho:_cons]*`gprime´
se of rho = .06130802
```

Alternatively one can use [R] **nlcom** to compute these standard errors.

```
. nlcom (   rho: tanh( _b[athrho:_cons]  ) ) ///
>       ( sigma: exp(  _b[lnsigma:_cons] ) )

         rho:  tanh( _b[athrho:_cons]  )
       sigma:  exp(  _b[lnsigma:_cons] )

------------------------------------------------------------------------------
        wage |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
         rho |   .6685435    .061308    10.90   0.000      .548382     .788705
       sigma |   6.095479   .1773995    34.36   0.000     5.747782    6.443175
------------------------------------------------------------------------------
```

Notice that the confidence intervals do not correspond with those in the output of `heckman`. This is because `heckman` first computes the bounds of the confidence intervals for the transformed parameters and then back-transforms those bounds to the original metric, while `nlcom` uses the standard errors for the back-transformed parameters for computing these bounds. In most cases computing the bounds on the transformed scale and then back-transforming those bounds to the original scale results in somewhat better bounds as the sampling distribution of the transformed parameters is likely to be better approximated by a normal distribution than the sampling distribution of the back-transformed parameters. For more discussion, see Sribney and Wiggins (2009). You can use the tricks discussed in Buis (2007) to recover the confidence intervals reported by `heckman`.

```
. di "confidence interval for rho: [" ///
>  tanh( _b[athrho:_cons] - invnormal(.975)*_se[athrho:_cons] ) ", " ///
```

```
>  tanh( _b[athrho:_cons] + invnormal(.975)*_se[athrho:_cons] ) "]"
confidence interval for rho: [.53049526, .77204703]
.
. di "confidence interval for sigma: [" ///
>  exp( _b[lnsigma:_cons] - invnormal(.975)*_se[lnsigma:_cons] ) ", " ///
>  exp( _b[lnsigma:_cons] + invnormal(.975)*_se[lnsigma:_cons] ) "]"
confidence interval for sigma: [5.7575126, 6.4532831]
```

Also note that `nlcom` returns the z-statistic and p-value for the test of the null hypothesis that the standard deviation and the correlation are zero, which were not reported by `heckman`. This test is problematic in case of the standard deviation, as this is a test 'on the boundary of the parameter space'. A standard deviation can only take values larger than or equal to zero, so the hypothesis that the standard deviation is equal to zero is on the boundary of the possible values for the standard deviation, and standard tests do not tend to behave well in this extreme area (for example: Gutierrez et al. 2001).

References

Buis, M. L. 2007. Stata tip 54: Where did my p-values go? *The Stata Journal* 7: 584–586.

———. 2011. Stata tip 97: Getting at ρs and σs. *The Stata Journal* 11: 315–317.

Feiveson, A. H. 2005. Explanation of the delta method. http://www.stata.com/support/faqs/statistics/delta-method/.

Gutierrez, R., S. Carter, and D. M. Drukker. 2001. On boundary-value likelihood-ratio tests. *Stata Technical Bulletin* 60: 15–18.

Sribney, W., and V. Wiggins. 2009. Standard errors, confidence intervals, and significance tests for ORs, HRs, IRRs, and RRRs. http://www.stata.com/support/faqs/statistics/delta-rule/.